Technical Report No. 208

GROMAX: A POTENTIAL PRODUCTIVITY ROUTINE FOR A TOTAL GRASSLAND ECOSYSTEM MODEL

David J. Connor[1/]

Natural Resource Ecology Laboratory

Colorado State University

Fort Collins, Colorado

GRASSLAND BIOME

U.S. International Biological Program

January 1973

[1/] On leave from La Trobe University, Bundoora, Victoria, Australia.

TABLE OF CONTENTS

ABSTRACT

GROMAX is a community photosynthesis model designed to describe the daily gross photosynthetic behavior of a mixed grassland community in response to daily radiation input. The community leaf structure is defined in terms of an area and angle combination for each of the component primary producers, and their intrinsic photosynthetic response is given in terms of either a measured or an idealized leaf photosynthesis-irradiance function. For a given community, gross daily carbon uptake is calculated for any site (latitude) and time (day of year) under clear sky conditions, overcast conditions, and by interpolation under the observed daily radiation.

INTRODUCTION

This model is written for inclusion in ELM and is designed to explain, on a daily basis, the fate of photosynthetically active radiation incident upon a grassland and the consequent upper limit to gross productivity. In operation, the predictions of this model are subject to modifications resulting from non-optimum conditions of the abiotic parameters of water availability, nutrient availability, and temperature, and to continuous losses from respiratory activity, grazing, and senescence. Although there is no way of validating this model directly, there is evidence from the literature dealing with the growth of ungrazed pastures and crops under agronomically optimum conditions which can be used to assess its general level of accuracy. In practice, the strict and continuous validation of any production-utilization-decomposition complex in a total grassland ecosystem model is not possible, and reliance must be placed upon coincidence of observed and simulated changes in biomass, together with any available fragmentary information on rates of individual processes.

Photosynthetic productivity is limited ultimately by the solar energy which is intercepted by vegetation and instantaneously coupled to the photochemical reduction of atmospheric carbon dioxide. An early model of community photosynthesis by Loomis and Williams (1963) used then available information on spectral composition of solar radiation and the quantum efficiency of the photosynthetic process to estimate potential productivity per unit of incident solar energy. A comparison between the predicted rate and the actual measured maximum rates of photosynthesis for a range of crops was an exciting assessment of our crop production capability and a challenge for subsequent crop modeling activities.

An important feature of the Loomis and Williams' model was its generality. Crop and environment were specified in such simple terms that its prediction could be applied with approximately equivalent accuracy (and, hence, also error) to many situations. This feature of generality seems to have become lost in the more advanced models of community photosynthesis which have been produced since that time. As the mechanisms that are portrayed become more exact, the definitions of environment and community required to drive the models become correspondingly more rigorous. Although, in theory, they may be appropriate to a range of communities, in practice they tend to become community-specific as a result of "shortcuts" or the availability of data.

In view of our general ignorance of detailed environmental and community parameters for the range of grassland communities, it is necessary that a generally applicable model be a simplified one. This does not mean that it cannot portray a process with a satisfying degree of realism, nor that it be structured in such a way that additional detail cannot be readily included as it becomes available. It does mean, however, that the level of approximation that must be used across all situations will most likely necessitate the omission of some detail which is currently available for isolated cases.

MODEL

Overall View

The major steps in the logic of the model are illustrated in Fig. 1. It is shown that for a defined locality and canopy structure, separate estimates are made of canopy illumination patterns and consequent canopy photosynthetic response for clear sky and overcast conditions, respectively. Canopy photosynthetic response under the observed radiation conditions is estimated by



Fig. 1. Sequence of calculations in GROMAX.

combining the responses for clear and overcast conditions on the basis of the estimated fractional cloudiness for the day. This figure is itself derived from the observed daily total radiation and the estimated daily radiation for clear and overcast conditions.

The model is designed to operate on a daily basis, and within each day the time step of integration is 1 hr. The model is constructed such that it need not be used in its entirety on each consecutive day in a continuous simulation. Since solar trace and canopy characteristics change little from day to day, it is intended that the complete routine be used once each 5 days or so, and that the ancillary estimates that it makes, viz.,

1. daily solar radiation under clear sky,
2. daily solar radiation under overcast sky,
3. daily canopy photosynthesis under clear sky, and
4. daily canopy photosynthesis under overcast sky,

be used with observed daily radiation values for the remaining days in the interval to estimate the canopy photosynthesis under changing radiation conditions.

The rerun of the entire model would be dictated by the need to accommodate an updated canopy structure or the need to adjust for the continuously changing solar trace. An interval of 5 days would accommodate the latter consideration.

Canopy Structure

The plant canopy is represented structurally by two parameters, green foliage area index (FAI) and foliage angle (FA), for each of the five possible producer components upon which ELM operates plus a single compartment for non-photosynthetic material (living plus dead) for all species.

These are listed together with the type species for the Pawnee Grassland in Table 1. For any particular run, only one photosynthetic component need be specified.

It is assumed that the dispersion and orientation of the individual foliage elements are at random. These assumptions are relevant to the geometrical treatments given to the penetration of radiation into the canopy and the resultant predictions of illumination patterns on the component foliage elements.

Radiation Climate (Subroutines TIME, SOLDEC, SOLALT, and GENRAD)

Locality definition consists of latitude (RLAT in degrees; by convention southern latitudes are negative), day of year (IDAY), and total daily solar radiation input (TOTRAD in cal/cm^2/day). From this, an hour-by-hour solar trace is calculated (TIME, SOLDEC, SOLALT). For each solar position the solar radiation appropriate to both clear and overcast sky is calculated (GENRAD) (de Wit, 1965). The observed daily solar radiation is used to estimate the fractional cloud cover (FOV) for the day by comparison with the estimates of daily radiation for clear sky and overcast sky conditions.

Canopy Illumination under Clear Sky Conditions (Subroutine ZZ)

For each hour (solar elevation), an analysis of the random gap and sunlit FAI is made by the method of Warren Wilson (1967). The distribution of light intensities on the canopy foliage elements depends upon irradiance above the canopy, solar angle, foliage angle, and foliage azimuth characteristics. Since it is assumed that the orientation (azimuth) of the foliage is at random, then the analysis of de Wit (1965) can be used to describe the distribution of irradiance upon them. This is achieved in this program by

Table 1. Canopy components.

Classification	Species on Pawnee Grassland
Warm-season grass	*Bouteloua gracilis*
Cool-season grass	*Agropyron smithii*
Forb	*Sphaeralcea coccinea*
Shrub	*Artemisia frigida*
Cactus	*Opuntia polyacantha*
Non-photosynthetic tissue	All species

assessing the distribution of irradiance on the sunlit foliage area for each hourly time step in 10 classes of incident angle (0.1 steps of the sine of the angle of incidence 0.05, 0.15, 0.25, ..., 0.95) between the solar beam and the foliage surfaces. The solutions to these calculations for 10° steps of solar angle-foliage angle are contained in the array ISINLS from Idso and de Wit (1970). (See BLOCK DATA in Appendix I).

The illumination pattern provided by direct beam radiation is supplemented by a diffuse radiation component, originating not only from sky radiation but also within the canopy by the scattering of intercepted direct beam radiation. The penetration of diffuse radiation is handled by the method of Anderson and Denmead (1969) (see next section), with an added component caused by the forward scattering of intercepted direct beam radiation in the canopy.

The canopy analysis is made in steps of 0.1 FAI, and the forward scattering coefficient for intercepted direct beam radiation (FSCAT) is taken to be 0.25 (de Wit, 1965).

Canopy Illumination under Overcast Conditions (Subroutine YY)

The geometrical considerations which govern the penetration of diffuse radiation into plant canopies are quite distinct from those which govern the penetration of direct beam radiation. A particular problem is that the equations which have been used to describe penetration of diffuse radiation have no analytical solution, so extensive numerical approximations are necessary. Anderson and Denmead (1969) have investigated this problem and have proposed a general approximation in the form of an exponential extinction profile with a leaf area dependent extinction coefficient. Their approximation is used here.

Calculation of Gross Photosynthetic Response

Empirical leaf photosynthesis-irradiance relationships for component species can be used to complete the step from canopy illumination patterns to a calculation of canopy photosynthesis. However, such data are generally not available, and so it is the object of this model to include an alternative procedure based upon a simplified representation of leaf photosynthetic response to irradiance.

Two parameters are used to define the photosynthetic response of each species component (I):

1. PSAT(I) the minimum irradiance (cal/cm^2/min, 0.4 to 0.7 μ) for maximum photosynthetic response and
2. QEFF(I) the average quantum efficiency (mole C/Einstein) of incident radiation below the saturation level.

These parameters are depicted in Fig. 2 with data from a comprehensive set of leaf response characteristics (Ludlow and Wilson, 1971). It can be seen that both PSAT and QEFF discriminate strongly between C_3 and C_4 photosynthetic types, and further that, although leaves of C_4 species do demonstrate a continued photosynthetic response up to irradiance and approaching full sunlight as many authors report, there is a large difference in photosynthetic efficiency at high and low irradiance. The form of the response is open to the type of simplification suggested, although clearly any established function is preferable and can be incorporated in the model.

Leaves of C_3 species show light saturation at comparatively low irradiance (0.08 to 0.15 cal/cm^2/min, 0.4 to 0.7 μ). In considering the implication of choosing an "arbitrary" value of PSAT for C_4 species, it is of interest to consider the levels of irradiance in canopies of these species under



Fig. 2. Leaf response functions for C_3 and C_4 species (from Ludlow and Wilson, 1971).

conditions of high insolation (say, 500 cal/cm^2/day). Since the FA in grass canopies is high (say, >55°), a very small proportion of foliage is ever subjected to irradiance over 0.40 cal/cm^2/min (0.4 to 0.7 μ). This is depicted in Fig. 3 where for a hypothetical canopy (FAI = 5, FA = 65°, QEFF = 0.05 mole C/Einstein), the influence of PSAT is considered by GROMAX under a range of solar radiation conditions. Even under insolation of 500 cal/cm^2/day, there is little response provided PSAT is greater than 0.30. This shows that there is little foliage illuminated with intensity greater than this value.

An important task of GROMAX is to functionally differentiate the photosynthetic potential of C_3, C_4, and mixed grassland types. In the absence of established leaf photosynthetic characteristics, this can be attempted with the simplifications depicted in Fig. 2, but for each species these approximations need to be replaced with actual measured response functions.

Ludlow and Wilson (1971) estimate mean maximum quantum efficiency values of 0.06 and 0.10 mole C/Einstein (absorbed) for C_3 and C_4 species, respectively. These values are consistent with determinations made on other species by several authors reviewed by Ludlow and Wilson. Loomis and Williams (1963) used a value of 0.10 mole C/Einstein of absorbed radiation in their assessment of potential productivity. Values of QEFF (average quantum efficiency of incident radiation up to PSAT) will always be less than these maximum rates. For example, values of QEFF for C_3 and C_4 species from Fig. 2 are 0.03 and 0.05 mole C/Einstein, respectively.

OUTPUT

Three attempts are made to demonstrate the general level of accuracy of GROMAX by comparing output with published data on productivity under



Fig. 3. Effect of the leaf light saturation characteristic (PSAT) on community photosynthesis.

optimum conditions. Assumptions are necessary at various points in the calculations and these are indicated. It should also be noted that while a prediction of gross photosynthesis is attempted, it is necessary to compare this figure with an observation on net photosynthesis. Depending upon what one's idea is of the magnitude of respiratory loss (50%?, 33%? of gross photosynthesis) and the problems in modeling this portion of the productivity picture, then reactions will vary as to the value of GROMAX as a practicable potential productivity routine. Perhaps one aspect of the problem can meet with general agreement. If the model were more complex and needed more parameters to drive it, then the chance of convincingly demonstrating its value with available data would be greatly diminished.

Photosynthesis of *Bouteloua gracilis* Grassland

The measured maximum daily photosynthesis of grassland dominated by blue grama on th Pawnee Grassland is ~4 g C/m^2/day (Dye, Brown, and Trlica, 1972). This was determined in June 1971 for a canopy with a FAI of 0.5 under insolation averaging 600 cal/cm^2/day.

Using

RLAT = 38°N

IDAY = 140

TOTRAD = 500 cal/cm^2/day

FAI = 0.5

and assuming

FA = 60°

QEFF = 0.06 mole C/Einstein of incident energy

PSAT = 0.40 cal/cm^2/min (0.4 to 0.7 μ),

estimated potential gross photosynthesis is 7 g C/m^2/day.

A second analysis which, by selection of appropriate parameter values, is appropriate to the photosynthesis of *B. gracilis* grassland is presented in Fig. 4. It is the relationship between photosynthesis and foliage area index for a range of radiation conditions. Observed foliage area indices at the Pawnee Site are less than 0.52, so that this figure demonstrates the restriction that low leaf area places upon potential photosynthesis.

Growth of a Wheat Crop

The data used here are taken from Paltridge et al. (1972) in a study which provides detailed crop and environmental records for a crop of wheat grown at Rutherglen, Victoria, Australia, in the winter-spring of 1971.

The measured growth rate (including roots) of this crop is recorded in Table 2, together with estimates of gross photosynthesis from GROMAX assuming

PSAT = 0.15 cal/cm^2/min (0.4 to 0.7 μ) and

QEFF = 0.04 mole C/Einstein incident radiation.

All other variables are available as measured parameters.

During the vegetative and early reproductive phases of growth when no serious moisture stress was recorded, GROMAX provides acceptable estimates of gross photosynthetic behavior.

Maximum Growth Rates of a Range of Communities

Peak growth rates for temperate and tropical communities have been reviewed by Begg (1965). For a range of sites, maximum recorded growth rate (net production), location (latitude), and insolation are provided. In order to utilize these data, a number of assumptions were necessary,



Fig. 4. Potential gross photosynthesis of *Bouteloua gracilis* prairie in relation to foliage area index.

Table 2. Growth of a wheat crop at Rutherglen, Australia.

Days from Sowing[a/]	FAI	Daily Radiation (cal/cm^2)	Growth ($g\ C/m^2/day$)	
			Observed Net Growth	Predicted Gross Uptake[b/]
25	0.1	197	0.3	0.5
50	0.6	199	0.8	2.1
75	1.8	240	1.8	3.6
100	4.1	255	3.9	4.8
125	6.0	439	5.6	7.2

a/ Sowing date: May 11, 1971.

b/ Assumed values in model: PSAT = 0.15 $cal/cm^2/min$ (0.4 to 0.7μ)
QEFF = 0.04 mole C/Einstein.

and these are tabulated together with the original data and GROMAX predictions in Table 3.

Unlike the previous example in which a trend, as well as actual values, can be used to gauge the value of the prediction, in this case a comparison is restricted to isolated data points. However, it can be seen that GROMAX does attempt to handle the C_3, C_4 problem and, in particular, that the reality of light saturation in C_3 species means that even under high radiation conditions and full canopy cover, the photosynthetically effective radiation in such canopies is much reduced.

CONCLUDING REMARKS

GROMAX is designed to calculate the potential gross photosynthesis of a pure or mixed-species grassland by functionally relating the component meteorological, ecological, and physiological processes. It provides a means of comparing the photosynthetic behavior of grassland sites and of making an initial step in the "analysis of the structure, function and utilization of grassland ecosystems" (Van Dyne, 1971). As a process model, it seeks to identify the intrinsic site, community, and species properties which control the overall process and, consequently, needs to be supported by field and laboratory process studies if it is to contribute to a deeper understanding of grassland productivity. At the present time, there is little leaf response data available for even the major species of the American grasslands, so that in the examples given approximation has been necessary. Even so, the importance of leaf response can be demonstrated by the important differences which exist primarily between photosynthetic pathway types but also within them.

Table 3. Measured peak growth rates, average insolation, and estimates of gross productivity (CIN).

Original Data[a]				Assumptions and Calculations					
Species and Country	RLAT	Peak Growth Rate (g C/m^2/day)	Average Insolation (cal/cm^2/day)	FAI	FA (°)	IDAY	PSAT (cal/cm^2/min, 0.4 to 0.7μ)	QEFF (mole C/Einstein)	CIN (g C/m^2/day)
Pennisetum typhoides Australia	14°S	20	510	5	65	10	0.40	0.08	24
Sorghum vulgare United States	39°N	19	690	5	65	140	0.40	0.07	28
Beta maritima England	52°N	11	294	5	30	140	0.15	0.06	11
Hordeum vulgare England	52°N	9	484	5	65	140	0.15	0.05	11
Trifolium sub-terraneum Australia	35°S	9	670	5	30	10	0.15	0.05	13
Grassica oleracea acephala England	52°N	8	382	5	30	140	0.15	0.05	10

a/ From Begg (1965).

LITERATURE CITED

Anderson, M. C., and O. T. Denmead. 1969. Short wave radiation on inclined surfaces in model plant communities. Agron. J. 61:867-872.

Begg, J. E. 1965. High photosynthetic efficiency in a low-latitude environment. Nature 205:1025-1026.

de Wit, C. T. 1965. Photosynthesis of leaf canopies. Agr. Res. Rep., Wageningen, Netherlands. 663 p.

Dye, A. J., L. F. Brown, and M. J. Trlica. 1972. Carbon dioxide exchange of blue grama as influenced by several ecological parameters, 1971. U.S. IBP Grassland Biome Tech. Rep. No. 181. Colorado State Univ., Fort Collins. 44 p.

Idso, S. B., and C. T. de Wit. 1970. Light relations in plant canopies. Appl. Optics 9:177-184.

Loomis, R. S., and W. A. Williams. 1963. Maximum crop productivity: An estimate. Crop Sci. 3:67-72.

Ludlow, M. M., and G. L. Wilson. 1971. Photosynthesis of tropical pasture plants. I. Illuminance, carbon dioxide concentration, leaf temperature, and leaf-air vapour pressure difference. Australian J. Biol. Sci. 24: 449-470.

Paltridge, G. W., A. C. Dilley, J. R. Garrat, G. I. Pearman, W. Shepherd, and D. J. Connor. 1972. The Rutherglen experiment on Sherpa wheat: Environmental and biological data. CSIRO, Australian Div. Atmos. Phys. Tech. Paper 22.

Van Dyne, G. M. [Principal Investigator]. 1971. Analysis of structure, function and utilization of grassland ecosystems: A progress report and a continuation proposal. U.S. IBP Grassland Biome, Colorado State Univ., Fort Collins. 363 p.

Warren Wilson, J. 1967. Stand structure and light penetration. III. Sunlit foliage area. J. Appl. Ecol. 4:159-165.

APPENDIX I

LISTING OF GROMAX

SUBROUTINE GROMAX CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.23.

```
      SUBROUTINE GROMAX(IDAY,RLAT,TOTRAD,FAI,FA,CIN,TOT,TOTOV,TOTCL,
     1CLEAR,CLOUDY,FOV)
C
C
C     **********************************************************************
C
C
C
C
C     TO RUN THIS SUBROUTINE SUPPLY THE FOLLOWING PARAMETERS
C
C
C     IDAY    JULIAN DAY
C
C     RLAT    LATITUDE IN DEGREES (SOUTHERN LATITUDES NEGATIVE)
C
C     TOTRAD  TOTAL DAILY SHORTWAVE RADIATION IN CAL/CM**2
C
C     FAI     A PREVIOUSLY DIMENSIONED ARRAY SIZE 6 CONTAINING THE
C             FOLIAGE AREA INDICES OF 5 PRODUCER COMPARTMENTS AND
C             NON PHOTOSYNTHETIC MATERIAL. SET NON ACTIVE COMPARTMENTS
C             TO ZERO
C
C     FA      A PREVIOUSLY DIMENSIONED ARRAY SIZE 6 CONTAINING THE
C             FOLIAGE ANGLE OF 5 PRODUCER COMPARTMENTS AND
C             NON PHOTOSYNTHETIC MATERIAL. SET NON ACTIVE COMPARTMENTS
C             TO ZERO
C
C
C     AND RECEIVE AS OUTPUT THE FOLLOWING PARAMETERS
C
C
C     CIN     INTO A PREVIOUSLY DIMENSIONED ARRAY SIZE 5 THE DAILY GROSS
C             PHOTOSYNTHESIS OF THE 5 PRODUCER COMPARTMENTS IN G CARBON/M**2
C
C     TOT     TOTAL DAILY GROSS PHOTOSYNTHESIS ALL COMPARTMENTS G CARBON/M**2
C
C     TOTOV   TOTAL DAILY GROSS PHOTOSYNTHESIS UNDER CLOUDY CONDITIONS
C
C     TOTCL   TOTAL DAILY GROSS PHOTOSYNTHESIS UNDER CLEAR CONDITIONS
C
C     CLEAR   DAILY SHORTWAVE RADIATION UNDER CLEAR CONDITIONS IN CAL/CM**2
C
C     CLOUDY  DAILY SHORTWAVE RADIATION UNDER CLOUDY CONDITIONS IN CAL/CM**2
C
C     FOV     FRACTION OF DAY WHICH WAS CLOUDY
C
C
C
C     **********************************************************************
C
C
      DIMENSION FAI(1),FA(1),CO(5),CC(5),CIN(1)
      DIMENSION BETA(12),DIRAD(12),DIFRAD(12),CLSK(12)
      CALL TIME(IDAY,RLAT,BETA)
      CALL GENRAD(TOTRAD,BETA,DIRAD,DIFRAD,CLSK,CLEAR,CLOUDY,FOV)
```

```
SUBROUTINE  GROMAX
                                   CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.23
      DO 19 I=1,5
      CIN(I)=0.
   19 CONTINUE
      TOT=0.
      TOTOV=0.
      TOTCL=0.
      DO 100 I=1,12
      IF(BETA(I))100,100,20
   20 CALL YY(DIFRAD(I),FAI,CO)
      CALL ZZ(DIRAD(I),CLSK(I),FAI,FA,BETA(I),CC)
      DO 90 J=1,5
      IF(FAI(J))90,90,80
   80 AV=(CO(J)*FOV+CC(J)*(1.-FOV))*2.
      CIN(J)=CIN(J)+AV
      TOT=TOT+AV
      TOTOV=TOTOV+CO(J)*2.
      TOTCL=TOTCL+CC(J)*2.
   90 CONTINUE
  100 CONTINUE
      RETURN
      END
```

SUBROUTINE ZZ   CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.23.

```
      SUBROUTINE ZZ(DIR,DIF,FAI,FA,BSUN,CARBON)
      COMMON/GEOM/ISINLS(9,9,10)
      DIMENSION FAI(1),FA(1),CARBON(1)
      DIMENSION PRAD(10),AREA(10),PHIB(6),SFAI(6),COVER(6)
      DATA SUBL/0.1/
      DATA FSCAT/0.25/
C     SUBL IS THE SIZE (FAI) OF EACH STEP IN THE CANOPY ANALYSIS
C     FSCAT IS THE FORWARD SCATTERING COEFFICIENT FOR DIRECT RADIATION
      TFAI=0.
      DO 10 I=1,5
      TFAI=TFAI+FAI(I)
      CARBON(I)=0.
   10 CONTINUE
      TFAI=TFAI+FAI(6)
      STEP=TFAI/SUBL
C     **************************************************************
C     THE FOLLOWING PARAMETERS DEFINE CONSTANT PROPERTIES FOR EACH SUBL
      TPHIB=0.
      TCOV=0.
      TSFAI=0.
      DO 100 J=1,6
      IF(FAI(J))100,100,50
   50 PHIB(J)=FAI(J)*PHI(FA(J),BSUN)/STEP
      TPHIB=TPHIB+PHIB(J)
      COVER(J)=1.-EXP(-PHIB(J))
      TCOV=TCOV+COVER(J)
      SFAI(J)=FAI(J)/PHIB(J)*COVER(J)/STEP
      TSFAI=TSFAI+SFAI(J)
  100 CONTINUE
      GAP=EXP(-TPHIB)
      SUNFAI=TFAI/TPHIB*(1.-GAP)/STEP
C     **************************************************************
      IB=IFIX(BSUN/10.)+1
      CGAP=1.
      CAPSUM=0.
      DIFR=DIF
      ISTEP=STEP
      CAP=0.
      DO 1000 NLAYER=1,ISTEP
      SUNFAI=SUNFAI*CGAP
C     SUNFAI IS NOW THE SUNLIT FOLIAGE AREA INDEX IN THE CURRENT SUBL
      DIFR=(DIFR+CAP*FSCAT)*ANDEN(SUBL)
C     DIFFUSE RADIATION INCLUDES AN EXTRA SCATTERED COMPONENT FROM
C     SUBL ABOVE THE CURRENT SUBL
      IF(DIFR.LE.0.)RETURN
      CGAP=CGAP*GAP
C     CGAP IS THE CUMULATIVE GAP TO A POINT IN THE CANOPY
      CAP=DIR*(1.-CGAP)-CAPSUM
C     CAP IS THE INTERCEPTED DIRECT BEAM RADIATION IN THE CURRENT SUBL
      CAPSUM=CAPSUM+CAP
      DO 800 J=1,5
      IF(FAI(J))800,800,130
  130 IF(SUNFAI.LT.0.000001)GO TO 600
      IA=IFIX(FA(J)/10.)+1
C     CALCULATE AREAS FOR IRRADIANCE CLASSES
```

SUBROUTINE ZZ

CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.

```
C     SPECIES CONTRIBUTE TO TOTAL SUNFAI IN PROPORTION TO THEIR INDIVIDUAL
C     SUNLIT FOLIAGE AREA INDEX CALCULATED AS ISOLATED SPECIES
      AREA(1)=FLOAT(ISINLS(IA,IB,1))/1000.*SUNFAI *SFAI(J)/TSFAI
      DO 250 K=2,10
      IF(ISINLS(IA,IB,K))230,230,150
  150 AREA(K)=FLOAT(ISINLS(IA,IB,K)-ISINLS(IA,IB,K-1))/1000.*SUNFAI*
     1SFAI(J)/TSFAI
      GO TO 250
  230 AREA(K)=0.
  250 CONTINUE
C     CALCULATE IRRADIANCE DISTRIBUTION ON LEAVES FROM THE RADIATION
C     INTERCEPTED BY THAT SPECIES DURING THIS HOUR INTERVAL
C     RADIATION INTERCEPTED BY SPECIES(J) IS PROPORTIONAL TO  ITS  CONTRIBUTION
C     TO TOTAL COVER
      CAPJ=CAP*COVER(J)/TCOV
      SUM=0.
C     THE BASIS OF THIS STEP IS THE 10 CLASSES OF SINLS 0.05,0.15,0.25,0.35,
C     SEE BLOCK DATA AND REFERENCE THERE
      DO 260 K=1,10
      IF(AREA(K))260,260,252
  252 SUM=SUM+AREA(K)*(FLOAT(K)/10.-0.05)
      IF(ISINLS(IA,IB,K).EQ.1)GO TO 265
  260 CONTINUE
  265 X=CAPJ/SUM
      DO 270 K=1,10
      PRAD(K)=X*(FLOAT(K)/10.-0.05)
  270 CONTINUE
C     PRAD IS IN CAL/CM**2/MIN FOR FUNCTION GPHS
      DO 500 K=1,10
      IF(AREA(K))500,500,400
  400 CARBON(J)=CARBON(J)+AREA(K)*GPHS(J,PRAD(K),DIFR)
  450 IF(ISINLS(IA,IB,K).EQ.1)GO TO 800
  500 CONTINUE
  600 CARBON(J)=CARBON(J)+(FAI(J)/STEP-SUNFAI*SFAI(J)/TSFAI)*
     1GPHS(J,DIFR)
  800 CONTINUE
 1000 CONTINUE
      RETURN
      END
```

```
SUBROUTINE  TIME
                                                   CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.
          SUBROUTINE TIME(IDAY,RLAT,BETA)
          DIMENSION BETA(1)
          DEC=SOLDEC(IDAY)
          X = 7.5
          DO 50 I=1,12
          IH=13-I
          BETA(IH)=SOLALT(X,DEC,RLAT)
          X=X+15.
       50 CONTINUE
          RETURN
          END
```

```
FUNCTION    S ____
                                                   CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.
          FUNCTION SOLDEC(I)
C         J.W. SPENCER SEARCH 5(2)
          T=2.*3.1416*FLOAT(I)/365.
          SOLDEC=0.6918E-2-0.399912*COS(T)+0.70257E-1*SIN(T)-0.6758E-2*COS(2
         1.*T)+0.907E-3*SIN(2.*T)-0.2697E-2*COS(3.*T)+0.148E-2*SIN(3.*T)
          RETURN
          END
```

```
FUNCTION    SOLALT
                                                   CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.2
          FUNCTION SOLALT(H,DEC,RLAT)
          DATA RADCON/0.017453/
          SINLAT=SIN(RLAT*RADCON)
          COSLAT=COS(RLAT*RADCON)
          A=ABS(H)*RADCON
          SINZ=COSLAT*COS(DEC)*COS(A)+SINLAT*SIN(DEC)
          IF(SINZ)15,15,8
        8 SOLALT=ATAN(SINZ/SQRT(1.-SINZ*SINZ))/RADCON
          RETURN
       15 SOLALT=0.
          RETURN
          END
```

SUBROUTINE GENRAD   CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.23

```
      SUBROUTINE GENRAD(TOTRAD,BETA,DIRAD,DIFRAD,CLSK,CLEAR,CLOUDY,FOV)
      DIMENSION BETA(1),DIRAD(1),DIFRAD(1),CLSK(1)
      DIMENSION SOLA(10),SUNY(10),DFCY(10),DFOY(10)
C     ARRAY SOLA CONTAINS SOLAR ANGLES IN DEGREES
C     SUNY,DFCY AND DFOY CONTAIN IN J/M**2/S THE CLEAR SKY DIRECT, THE CLEAR
C     SKY DIFFUSE AND THE OVERCAST SKY DIFFUSE IRRADIANCE CORRESPONDING TO THE
C     SOLAR ANGLES IN ARRAY SOLA
      DATA SOLA/0.,5.,15.,25.,35.,45.,55.,65.,75.,90./
      DATA SUNY/0.,29.3,88.,175.,262.,336.,402.,452.,483.,504./
      DATA DFCY/0.,29.,42.,49.,56.,64.,68.,71.,75.,77./
      DATA DFOY/0.,6.,26.,45.,64.,80.,94.,105.,112.,116./
      CLEAR=0.
      CLOUDY=0.
      DO 10 I=1,12
      SA=BETA(I)
      IF(SA)8,8,6
    6 DIRAD(I)= ALINT(SOLA,SUNY,10,SA)*2.*0.08596
      CLSK(I)=ALINT(SOLA,DFCY,10,SA)*1.7*0.08596
      DIFRAD(I)=ALINT(SOLA,DFOY,10,SA)*1.7*0.08596
      CLEAR=CLEAR+(DIRAD(I)+CLSK(I))*2.
      DIRAD(I)=DIRAD(I)/60.
      CLSK(I)=CLSK(I)/60.
      CLOUDY=CLOUDY+DIFRAD(I)*2.
      DIFRAD(I)=DIFRAD(I)/60.
      GO TO 10
    8 DIRAD(I)=0.
      CLSK(I)=0.
      DIFRAD(I)=0.
   10 CONTINUE
      FOV=(CLEAR-TOTRAD )/(CLEAR-CLOUDY)
      IF(FOV.GT.1.)FOV=1.
      IF(FOV.LT.0.)FOV=0.
      RETURN
      END
```

FUNCTION ALINT   CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.23.

```
      FUNCTION ALINT(TABX,TABY,N,XVAL)
C     LINEAR INTERPOLATION ROUTINE
C     TABX MUST BE A CONTINUOUSLY ASCENDING ARRAY
C     IF XVAL IS OUT OF RANGE OF TABX THE FIRST OR LAST ENTRY IN TABY WILL
C     BE RETURNED
      DIMENSION TABX(1),TABY(1)
      IF(XVAL.LE.TABX(1))GO TO 20
      IF(XVAL.GE.TABX(N))GO TO30
      DO 10 I=2,N
      IF(XVAL.LE.TABX(I))GO TO 15
   10 CONTINUE
   15 CONTINUE
      ALINT=(XVAL-TABX(I-1))*(TABY(I)-TABY(I-1))/(TABX(I)-TABX(I-1))
     1+TABY(I-1)
      RETURN
   20 ALINT=TABY(1)
      RETURN
   30 ALINT=TABY(N)
      RETURN
      END
```

```
SUBROUTINE  YY                                CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.23.

          SUBROUTINE YY(DIFRAD,FAI,CO)
          DIMENSION FAI(1),CO(1)
          DATA SUBL/0.1/
          TFAI=0.
          DO 10 I=1,6
          TFAI=TFAI+FAI(I)
          IF(I.EQ.6)GO TO 10
          CO(I)=0.
       10 CONTINUE
          STEP=TFAI/SUBL
          ISTEP=STEP
          CA=0.
          DO 100 I=1,ISTEP
          CA=CA+SUBL
          DIF=DIFRAD*ANDEN(CA)
          IF(DIF.LE.0.)RETURN
          DO 50 J=1,5
          IF(FAI(J))50,50,20
       20 CO(J)=CO(J)+FAI(J)/STEP*GPHS(J,DIF)
       50 CONTINUE
      100 CONTINUE
          RETURN
          END

                                              CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.23.
C         FUNCTION ANDEN
          FUNCTION ANDEN(A)
C         ANDERSON AND DENMEAD TREATMENT OF THE PENETRATION OF DIFFUSE
C         RADIATION FROM AGRON.JOUR. 61(6),1969
          IF(A-0.5)10,10,20
       10 ANDEN=1./EXP(1.25*A)
          RETURN
       20 IF(A-1.0)30,30,40
       30 ANDEN=1./EXP(1.1*A)
          RETURN
       40 ANDEN=1./EXP(0.9*A)
          RETURN
          END

FUNCTION  GPHS                                CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.23.

          FUNCTION GPHS(ISP,SWRAD)
          DIMENSION PSAT(5),QEFF(5)
          DATA QEFF/0.05,4*0.03/
          DATA PSAT/0.40,4*0.15/
C         ALL LEAF RESPONSE DATA ARE BASED UPON PHAR
C         GPHS WILL USE EITHER AN EMPIRICAL        PHOTOSYNTHESIS - PHAR (CAL/CM**2/
C         MIN) RESPONSE FUNCTION IF IT IS AVAILABLE FOR ANY SPECIES OR ALTERNATIVELY
C         USE A SIMPLIFIED PSAT/QEFF RELATIONSHIP. THE EMPIRICAL FUNCTION WILL BE
C         USED IF THE CORRESPONDING PSAT/QEFF COMBINATION IS SET TO ZERO
          IF(PSAT(ISP))95,95,5
        5 IF(SWRAD*0.45.GT.PSAT(ISP))SWRAD=PSAT(ISP)/0.45
C         GPHS=SWRAD*QEFF(ISP)*12.*60./100.*8.64
          GPHS=SWRAD*QEFF(ISP)*62.21
C         THIS IS IN GRAM CARBON/M**2/HR WHICH IS THE TIME STEP
C         SEE LOOMIS AND WILLIAMS FOR THIS CONVERSION (CROP SCI. 3 1963 P67-72)
          RETURN
       95 CONTINUE
C         CODE THE EMPIRICAL RESPONSE FUNCTIONS IN 100,200,300,400,500
C         GO TO(100,200,300,400,500),ISP
          RETURN
          END


FUNCTION  PHI                                 CDC 6400 FTN V3.0-P308 OPT=1  01/11/73  09.15.23.

          FUNCTION PHI(A,B)
C         SEE WARREN-WILSON(1967) J. APPL. ECOL. 4 P159-65.
          DATA RADCON/0.017453/
          IF(A-B)10,10,20
       10 PHI=COS(A*RADCON)
          RETURN
       20 IF(A-90.)30,40,40
       30 Z=SIN(B*RADCON)*COS(A*RADCON)/COS(B*RADCON)/SIN(A*RADCON)
          Z=ATAN(SQRT(1.-Z*Z)/Z)
          PHI=COS(A*RADCON)*(1.+2./3.1416*(SIN(Z)/COS(Z)-Z))
          RETURN
       40 PHI=2./3.1416*COS(B*RADCON)/SIN(B*RADCON)
          RETURN
          END
```

BLOCK DATA

CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.2

```
      BLOCK DATA
C     SEE IDSO AND DE WIT (1970) J. APPL. OPTICS 9 P 177-84.
      COMMON/GEOM/ISINLS(9,9,10)
      DATA (((ISINLS(I,J,K),I=1,9),J=1,9),K=1,2)/
     1 234,  81,  29,  15,  10,   8,   6,   5,   5,   0,  41,  33,  17,
     1  11,   8,   7,   6,   5,   0,   0,  21,  22,  13,   9,   7,   7,
     1   6,   0,   0,   0,  15,  18,  12,   9,   8,   8,   0,   0,   0,
     1   0,  14,  18,  13,  11,  10,   0,   0,   0,   0,   0,  15,  22,
     1  17,  15,   0,   0,   0,   0,   0,   0,  21,  33,  29,   0,   0,
     1   0,   0,   0,   0,   0,  41,  81,   0,   0,   0,   0,   0,   0,
     1   0,   0, 234,1000, 405, 122,  63,  41,  31,  25,  22,  21, 183,
     1 124, 168,  70,  44,  33,  26,  23,  22,   0,  68,  62, 108,  53,
     1  38,  30,  26,  25,   0,   0,  44,  45,  90,  49,  38,  33,  31,
     1   0,   0,   0,  36,  41,  90,  53,  44,  41,   0,   0,   0,   0,
     1  36,  45, 108,  70,  63,   0,   0,   0,   0,   0,  44,  62, 168,
     1 122,   0,   0,   0,   0,   0,   0,  68, 124, 405,   0,   0,   0,
     1   0,   0,   0,   0, 183,1000/
      DATA (((ISINLS(I,J,K),I=1,9),J=1,9),K=3,4)/
     1
     1                             0, 651, 316, 150,  95,  70,  57,
     1  50,  47, 577, 252, 232, 181, 104,  75,  61,  53,  50,   0, 185,
     1 120, 147, 137,  88,  70,  61,  57,   0,   0, 117,  85, 120, 125,
     1  88,  75,  70,   0,   0,   0,  97,  77, 120, 137, 104,  95,   0,
     1   0,   0,   0,  97,  85, 147, 181, 150,   0,   0,   0,   0,   0,
     1 117, 120, 232, 316,   0,   0,   0,   0,   0,   0, 185, 252, 651,
     1   0,   0,   0,   0,   0,   0,   0, 577,   0,   0,1000, 617, 292,
     1 178, 129, 104,  91,  85,1000, 450, 326, 344, 201, 139, 111,  97,
     1  91, 357, 306, 196, 199, 259, 170, 129, 111, 104,   0, 157, 189,
     1 138, 162, 237, 170, 139, 129,   0,   0, 114, 154, 124, 162, 259,
     1 201, 178,   0,   0,   0, 103, 154, 138, 199, 344, 292,   0,   0,
     1   0,   0, 114, 189, 196, 326, 617,   0,   0,   0,   0,   0, 157,
     1 306, 450,1000,   0,   0,   0,   0,   0,   0, 357,1000,   0/
      DATA (((ISINLS(I,J,K),I=1,9),J=1,9),K=5,6)/
     1
     1   0, 999, 626, 299, 210, 167, 146, 136,   0, 998, 463, 420, 431,
     1 232, 181, 156, 146, 998, 460, 296, 269, 310, 366, 215, 181, 167,
     1   0, 300, 270, 203, 216, 283, 366, 232, 210,   0,   0, 213, 219,
     1 182, 216, 310, 431, 299,   0,   0,   0, 193, 219, 203, 269, 420,
     1 626,   0,   0,   0,   0, 213, 270, 296, 464,1000,   0,   0,   0,
     1   0,   0, 300, 460,1000,   0,   0,   0,   0,   0,   0,   0,1000,
     1   0,   0,   0,   0,1000, 787, 493, 324, 253, 218, 203,   0,1000,
     1 696, 526, 500, 379, 277, 235, 218,1000, 699, 429, 359, 376, 419,
     1 350, 277, 253, 595, 451, 368, 283, 283, 340, 419, 379, 324,   0,
     1 274, 309, 294, 252, 283, 376, 500, 493,   0,   0, 216, 278, 294,
     1 283, 359, 526, 788,   0,   0,   0, 216, 309, 368, 429, 696,   0,
     1   0,   0,   0,   0, 274, 451, 699,   0,   0,   0,   0,   0,   0,
     1   0, 595,   0,   0,   0/
      DATA (((ISINLS(I,J,K),I=1,9),J=1,9),K=7,8)/
     1
     1                             0,   0,   0,1000, 770, 494, 370, 314,
     1 291,   0,   0,1000, 692, 593, 574, 418, 342, 314,   0,1000, 632,
     1 480, 461, 486, 529, 418, 370,1000, 653, 493, 385, 368, 414, 486,
     1 574, 494, 449, 445, 418, 383, 339, 368, 461, 593, 770,   0, 242,
     1 342, 373, 383, 385, 480, 692,1000,   0,   0, 211, 342, 418, 493,
     1 632,1000,   0,   0,   0,   0, 242, 445, 653,1000,   0,   0,   0,
     1   0,   0,   0, 449,1000,   0,   0,   0,   0,   0,   0,   0,1000,
     1 791, 542, 445, 408,   0,   0,   0,1000, 736, 666, 654, 497, 445,
```

BLOCK DATA   CDC 6400 FTN V3.0-P308 OPT=1 01/11/73 09.15.23.

```
      1   0.   0,1000, 664, 578, 575, 605, 654, 542,   0,1000, 676, 522,
      1 480, 510, 575, 666, 791,1000, 655, 553, 496, 450, 480, 578, 736,
      11000, 378, 459, 475, 485, 496, 522, 664,1000,   0,   0, 229, 393,
      1 475, 553, 676,1000,   0,   0,   0,   0, 229, 459, 655,1000,   0,
      1   0,   0,   0,   0,   0, 378,1000,   0,   0,   0,   0/
       DATA (((ISINLS(I,J,K),I=1,9),J=1,9),K=9,10)/
      1                                                           0,   0,
      1   0,   0,   0,1000, 847, 651, 579,   0,   0,   0,   0,1000, 817,
      1 759, 759, 651,   0,   0,   0,1000, 770, 705, 713, 759, 847,   0,
      1   0,1000, 740, 640, 645, 705, 817,1000,   0,1000, 760, 654, 604,
      1 640, 770,1000,   0,1000, 721, 646, 633, 654, 740,1000,   0,   0,
      1 462, 533, 583, 646, 760,1000,   0,   0,   0,   0, 316, 533, 721,
      11000,   0,   0,   0,   0,   0,   0, 462,1000,   0,   0,   0,   0,
      1   0,   0,   0,   0,   0,   0,   0,1000,1000,1000,   0,   0,   0,
      1   0,   0,1000,1000,1000,1000,   0,   0,   0,   0,1000,1000,1000,
      11000,1000,   0,   0,   0,1000,1000,1000,1000,1000,   0,   0,   0,
      11000,1000,1000,1000,1000,   0,   0,   0,1000,1000,1000,1000,1000,
      1   0,   0,   0,1000,1000,1000,1000,1000,   0,   0,   0,   0,1000,
      11000,1000,1000,   0,   0,   0,   0,   0,1000,1000,1000,   0,   0,
      1   0,   0,   0,   0/
       END
```

```
  01/11/73  CSU SCOPE 3.3  B    C012 C013 C140 C141 12/11/72
09.15.21.TA2923M
09.15.21.TA292,A1F0****,T20.CONNOR
09.15.21.FTN(R=0)
09.15.59.      4.804 CP SECONDS COMPILATION TIME
09.15.59.REWIND(OUTPUT)
09.15.59.COPYCF(OUTPUT,D)
09.15.59.FL= 010000 CP 00004.812SEC.  IO 00024.705SEC.
09.16.01.EOF/EOI ENCOUNTERED
09.16.01.FL= 001000 CP 00004.822SEC.  IO 00025.155SEC.
09.16.07.FL= 043000 CP 00004.822SEC.  IO 00025.155SEC.
09.16.07.REWIND(D)
09.16.07.COPYCF(D,OUTPUT)
09.16.07.FL= 010000 CP 00004.825SEC.  IO 00025.195SEC.
09.16.08.FL= 001000 CP 00004.826SEC.  IO 00025.650SEC.
09.16.08.FL= 043000 CP 00004.827SEC.  IO 00025.650SEC.
09.16.08.CP        4.827 SEC.
09.16.08.IO       25.650 SEC.
```